Имеется справочник "Клиенты" с табличной частью "Автомобиль" с такими полями как "Наименование", "НомерДоговора", "Количество". Имеется документ "ОказаниеУслуг" с полями на форме "Клиент" (СправочникСсылка.Клиенты) и "Автомобиль" (СправочникСсылка.Автомобили). Необходимо при выборе клиента из справочника в поле "Автомобиль" выводить список Автомобилей принадлежащих данному клиенту для дальнейшего выбора.
---
```1c
    //СЛУЖЕБНАЯ ПРОЦЕДУРА
    //Получаем список автомобилей клиента и загружаем их в список выбора
    &НаСервере
    Процедура СформироватьСписокАвтомобилейКлиентаНаСервере()

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | КлиентыАвтомобили.Автомобиль КАК Автомобиль
    |ИЗ
    | Справочник.Клиенты.Автомобили КАК КлиентыАвтомобили
    |ГДЕ
    | КлиентыАвтомобили.Ссылка = &КлиентСсылка";

    Запрос.УстановитьПараметр("КлиентСсылка", Объект.Клиент);

    РезультатЗапроса = Запрос.Выполнить().Выгрузить();

    массАвтомобилейКлиента = РезультатЗапроса.ВыгрузитьКолонку("Автомобиль");

    //Очищаем переменную в объекте
    Объект.Автомобиль = Справочники.Автомобили.ПустаяСсылка();

    //Очищаем список выбора и загружаем полученные значения
    Элементы.Автомобиль.СписокВыбора.Очистить();
    Элементы.Автомобиль.СписокВыбора.ЗагрузитьЗначения(массАвтомобилейКлиента);

    КонецПроцедуры

    //ОБРАБОТЧИКИ СОБЫТИЙ ЭЛЕМЕНТОВ ФОРМЫ
    &НаСервере
    Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)
    //Заполняем список значениями по клиенту
    Если ЗначениеЗаполнено(Объект.Клиент) Тогда
    СформироватьСписокАвтомобилейКлиентаНаСервере();
    КонецЕсли;
    КонецПроцедуры

    &НаКлиенте
    Процедура КлиентПриИзменении(Элемент)
    //При изменении клиента получаем его список автомобилей
    СформироватьСписокАвтомобилейКлиентаНаСервере();
    КонецПроцедуры
```

у реквизита формы Автомобили, в палитре свойств поставить флажок "Режим выбора из списка"
---
В обработчик При изменении реквизита Клиент, запросом в спр Клиенты дергайте тч с авто и формируйте список выбора.
Либо через выбранное значение в табличную часть лезть и формировать список выбора.
---
Можно примерный код для этого?
---
```1c
    //СЛУЖЕБНАЯ ПРОЦЕДУРА
    //Получаем список автомобилей клиента и загружаем их в список выбора
    &НаСервере
    Процедура СформироватьСписокАвтомобилейКлиентаНаСервере()

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | КлиентыАвтомобили.Автомобиль КАК Автомобиль
    |ИЗ
    | Справочник.Клиенты.Автомобили КАК КлиентыАвтомобили
    |ГДЕ
    | КлиентыАвтомобили.Ссылка = &КлиентСсылка";

    Запрос.УстановитьПараметр("КлиентСсылка", Объект.Клиент);

    РезультатЗапроса = Запрос.Выполнить().Выгрузить();

    массАвтомобилейКлиента = РезультатЗапроса.ВыгрузитьКолонку("Автомобиль");

    //Очищаем переменную в объекте
    Объект.Автомобиль = Справочники.Автомобили.ПустаяСсылка();

    //Очищаем список выбора и загружаем полученные значения
    Элементы.Автомобиль.СписокВыбора.Очистить();
    Элементы.Автомобиль.СписокВыбора.ЗагрузитьЗначения(массАвтомобилейКлиента);

    КонецПроцедуры

    //ОБРАБОТЧИКИ СОБЫТИЙ ЭЛЕМЕНТОВ ФОРМЫ
    &НаСервере
    Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)
    //Заполняем список значениями по клиенту
    Если ЗначениеЗаполнено(Объект.Клиент) Тогда
    СформироватьСписокАвтомобилейКлиентаНаСервере();
    КонецЕсли;
    КонецПроцедуры

    &НаКлиенте
    Процедура КлиентПриИзменении(Элемент)
    //При изменении клиента получаем его список автомобилей
    СформироватьСписокАвтомобилейКлиентаНаСервере();
    КонецПроцедуры
```

у реквизита формы Автомобили, в палитре свойств поставить флажок "Режим выбора из списка"
---
Как я поняла это всё нужно писать в модуле форму документа?
Написала вот так, но список выбора не открывается никакой
---
Извините, всё работает. Просто не указала в свойствах события для формы и клиента.
Спасибо большое за помощь